# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Bartłomiej Piotrowski <bpiotrowski@archlinux.org>
# Contributor: yugrotavele <yugrotavele at archlinux dot us>
# Contributor: Arkham <arkham at archlinux dot us>
# Contributor: Christoph Zeiler <archNOSPAM_at_moonblade.dot.org>
# Contributor: Arnaud Durand <biginozNOSPAM_at_free.dot.fr>
# Contributor: Nuno Araujo <nuno.araujo at russo79.com>

pkgname=bleachbit
pkgver=4.6.0
pkgrel=1
pkgdesc='Deletes unneeded files to free disk space and maintain privacy'
arch=(any)
url='https://www.bleachbit.org/'
license=(GPL3)
depends=(gtk3
         python
         python-gobject)
makedepends=(git)
source=(git+https://github.com/bleachbit/bleachbit.git#tag=v${pkgver}?signed)
validpgpkeys=(A9E582E4054A159315EDC943D6D447B02B4D4C9D) # Andrew Ziem <ahz001@gmail.com>
sha256sums=('SKIP')

package() {
  cd $pkgname
  make prefix=/usr DESTDIR="$pkgdir" install
}
